

# Computerpraktikum

Block I

## Die erste und die zweite Cohomologiegruppe

Matthias Künzer

Universität Stuttgart

12. Mai 2019

## Inhalt

1	Magma	3
2	Derivationen und $H^1$	4
3	Umgang mit Gruppen	9
4	Gruppen in Erzeugern und Relationen	10
5	Berechnung von $H^1$ für Gruppen in Erzeugern und Relationen	12
6	2-Cozykel und $H^2$	27
7	Zusammenhang von $H^2$ und $H^1$	28
8	Berechnung von $H^2$ für Gruppen in Erzeugern und Relationen	28

# Vorwort

Sei  $G$  eine endliche Gruppe.

Sei  $M$  ein  $G$ -Modul. D.h. sei  $M$  eine abelsche Gruppe, zusammen mit einer Abbildung  $G \times M \rightarrow M$ , die Eigenschaften einer äußeren Operation von  $G$  auf  $M$  erfüllt.

Z.B. können wir  $M = \mathbb{Z}$  wählen, zusammen mit der Abbildung  $G \times \mathbb{Z} \rightarrow \mathbb{Z}$ , für welche  $g \cdot z = z$  ist für  $g \in G$  und  $z \in \mathbb{Z}$ .

Eine Derivation sei eine Abbildung  $d$  von  $G$  nach  $M$ , für die stets  $d(g \cdot g') = d(g) + g \cdot d(g')$  ist.

Eine innere Derivation sei eine solche Abbildung  $d$  der Form  $d(g) = g \cdot m - m$  für ein festes  $m \in M$ .

Man definiert nun die erste Cohomologiegruppe  $H^1(G, M)$  als die Gruppe der Derivationen modulo der Untergruppe der inneren Derivationen.

Ähnlich, nur etwas aufwendiger dann die zweite Cohomologiegruppe.

Wir wollen die erste und die zweite Cohomologiegruppe berechnen mittels eines Verfahrens, das auf ZASSENHAUS [6] zurückgeht, von HOLT, EICK und O'BRIEN [2, §7.6] in den Cohomologie-Kontext übersetzt wurde und welches ich von CHRISTIAN WEBER [4, §1.5], [5, §5] kenne.

Dank geht an LUKAS HAUGER, FELICITAS WALTER und JULIEN FLAD für Korrekturen.

Für weitere Hinweise auf Fehler und Unklarheiten bin ich dankbar.

Stuttgart, Frühjahr 2019

Matthias Künzer

## 1 Magma

Wir verwenden das Computeralgebra-System Magma [1].

Das Handbuch findet sich unter `magma.maths.usyd.edu.au/magma/`.

Es kann Magma aufgerufen werden durch Eingabe von `magma` in einer Shell auf einem stud-Rechner.

Man teste einmal `1+1`; und dann `Enter`.

## 2 Derivationen und $H^1$

Sei  $G$  eine Gruppe.

**Definition 1** Ein  $G$ -Modul ist eine abelsche Gruppe  $M$ , zusammen mit einer Abbildung

$$\begin{aligned} G \times M &\rightarrow M \\ (g, m) &\mapsto g \cdot m \end{aligned}$$

für welche

$$\begin{aligned} g \cdot (m + m') &= g \cdot m + g \cdot m' \\ 1 \cdot m &= m \\ g \cdot (g' \cdot m) &= (g \cdot g') \cdot m \end{aligned}$$

gelten für  $g, g' \in G$  und  $m, m' \in M$ .

**Beispiel 2** Sei  $k \in \mathbb{Z}$ . Sei  $M = \mathbb{Z}/k\mathbb{Z}$ . Sei  $g \cdot m := m$  für  $g \in G$  und  $m \in M$ .

Man spricht vom *trivialen*  $G$ -Modul  $\mathbb{Z}/k\mathbb{Z}$ .

Für  $k > 0$  prim werden wir dieses Beispiel für erste direkte Berechnungen verwenden können, da dann alle betrachteten Mengen endlich werden.

**Beispiel 3** Sei  $G$  endlich.

Der *Gruppenring*  $\mathbb{Z}G$  hat als additive Gruppe den freien  $\mathbb{Z}$ -Modul auf der Menge  $G$ .

Es besteht also  $\mathbb{Z}G$  aus den formalen  $\mathbb{Z}$ -Linearkombinationen der Elemente aus  $G$ , i.e.

$$\mathbb{Z}G = \left\{ \sum_{g \in G} z_g g : z_g \in \mathbb{Z} \text{ für } g \in G \right\}.$$

Die Multiplikation auf dem Ring  $\mathbb{Z}G$  ist gegeben durch die Multiplikation auf  $G$ , distributiv fortgesetzt.

Also ist

$$\left( \sum_{g \in G} z_g g \right) \cdot \left( \sum_{g \in G} z'_g g \right) = \sum_{g, h \in G} z_g z'_h g \cdot h,$$

wobei  $z_g, z'_g \in \mathbb{Z}$  und wobei  $g \cdot h$  in  $G$  gebildet wird.

Mit anderen Worten,  $\mathbb{Z}G$  trägt die Multiplikation von  $G$ , distributiv fortgesetzt.

Es ist  $\mathbb{Z}G$  ein  $G$ -Modul, mit der Multiplikation

$$h \cdot \sum_{g \in G} z_g g := \sum_{g \in G} z_g h \cdot g$$

für  $h \in G$  und  $\sum_{g \in G} z_g g \in \mathbb{Z}G$ , i.e. mit der vom Ring  $\mathbb{Z}G$  stammenden Multiplikation. Dieser Modul heißt auch *regulärer*  $G$ -Modul.

**Bemerkung 4** Sei  $d \geq 1$ . Sei  $M := \mathbb{Z}^{d \times 1}$ . Sei

$$\gamma : G \rightarrow \mathbb{Z}^{d \times d}$$

eine Abbildung, für welche  $\gamma(1) = E_d$  und  $\gamma(g \cdot g') = \gamma(g) \cdot \gamma(g')$  gilt.

Dann wird  $M$  zu einem  $G$ -Modul, wenn wir

$$g \cdot m := \gamma(g) \cdot m$$

setzen für  $g \in G$  und  $m \in M$ , wobei letztere Multiplikation eine Matrix  $\gamma(g) \in \mathbb{Z}^{d \times d}$  mit einem Vektor  $m \in M = \mathbb{Z}^{d \times 1}$  multipliziert.

**Definition 5** Sei  $M$  ein  $G$ -Modul.

Eine *Derivation* von  $G$  in  $M$  ist eine Abbildung

$$\delta : G \rightarrow M$$

welche

$$\delta(g \cdot g') = \delta(g) + g \cdot \delta(g')$$

erfüllt für  $g, g' \in G$ .

Wir schreiben

$$Z^1(G, M)$$

für die Menge der Derivationen von  $G$  in  $M$ .

Da die Nullabbildung, die jedes Element von  $G$  auf  $0 \in M$  schickt, eine Derivation ist, und da für Derivationen  $\delta$  und  $\tilde{\delta}$  auch  $\delta - \tilde{\delta}$  eine Derivation ist, ist  $Z^1(G, M)$  eine Untergruppe der abelschen Gruppe aller Abbildungen von  $G$  nach  $M$ .

Insbesondere ist  $Z^1(G, M)$  eine abelsche Gruppe.

**Bemerkung 6** Sei  $\delta \in Z^1(G, M)$ .

Es ist

$$0 = \delta(1) - \delta(1) = \delta(1 \cdot 1) - \delta(1) = \delta(1) + 1 \cdot \delta(1) - \delta(1) = \delta(1) .$$

Für  $g \in G$  ist  $0 = \delta(1) = \delta(g \cdot g^{-1}) = \delta(g) + g \cdot \delta(g^{-1})$  und also

$$\delta(g^{-1}) = -g^{-1} \cdot \delta(g) .$$

Für  $m \geq 1$  und  $g_1, g_2, \dots, g_m \in G$  erhalten wir

$$\delta(g_1 \cdot g_2 \cdot \dots \cdot g_m) = \delta(g_1) + g_1 \cdot \delta(g_2) + g_1 \cdot g_2 \cdot \delta(g_3) + \dots + g_1 \cdot g_2 \cdot \dots \cdot g_{m-1} \cdot \delta(g_m)$$

**Definition 7** Sei  $m \in M$ .

Sei die Abbildung  $\delta_m^{\text{inn}} : G \rightarrow M$  definiert durch

$$\delta_m^{\text{inn}}(g) := g \cdot m - m$$

für  $g \in G$ .

Es ist  $\delta_m^{\text{inn}}$  eine Derivation, da

$$\delta_m^{\text{inn}}(g) + g \cdot \delta_m^{\text{inn}}(g') = m - g \cdot m + g \cdot (m - g' \cdot m) = m - g \cdot g' \cdot m = \delta_m^{\text{inn}}(g \cdot g')$$

ist für  $g, g' \in G$ . Eine Derivation der Form  $\delta_m^{\text{inn}}$  für ein  $m$  heißt auch *innere Derivation*.

Wir schreiben

$$B^1(G, M) := \{ \delta_m^{\text{inn}} : m \in M \} \subseteq Z^1(G, M)$$

für die Menge der inneren Derivationen.

Es ist  $\delta_0^{\text{inn}} = 0$  und  $\delta_m^{\text{inn}} - \delta_{m'}^{\text{inn}} = \delta_{m-m'}^{\text{inn}}$ . Folglich ist  $B^1(G, M)$  eine Untergruppe der abelschen Gruppe  $Z^1(G, M)$ .

**Definition 8** Die Faktorgruppe

$$H^1(G, M) := Z^1(G, M) / B^1(G, M)$$

heißt *erste Cohomologiegruppe* von  $G$  in  $M$ .

Ihre Elemente sind also von der Form  $\delta + B^1(G, M)$  mit  $\delta \in Z^1(G, M)$ .

Hierbei gilt  $\delta + B^1(G, M) = \delta' + B^1(G, M)$  genau dann, wenn  $\delta - \delta' \in B^1(G, M)$  liegt, wobei  $\delta, \delta' \in Z^1(G, M)$ .

Ferner wird in der abelschen Gruppe  $H^1(G, M)$  repräsentantenweise addiert, i.e.

$$(\delta + B^1(G, M)) + (\delta' + B^1(G, M)) = (\delta + \delta') + B^1(G, M),$$

wobei  $\delta, \delta' \in Z^1(G, M)$ .

Uns soll weniger interessieren, wozu man  $H^1(G, M)$  ausrechnet, als wie man diese Gruppe ausrechnet.

Wer es dennoch wissen will, der findet Antworten e.g. in [3, §3.1] oder in der Algebraischen Topologie.

**Bemerkung 9** Ist  $G$  endlich, so ist

$$|G| \cdot H^1(G, M) = 0$$

aus der Theorie bekannt.

Ist  $M$  endlich und  $\exp(M)$  das kleinste gemeinsame Vielfache der Ordnungen der Elemente von  $M$ , so ist  $\exp(M) \cdot H^1(G, M) = 0$ . Insbesondere ist dann

$$|M| \cdot H^1(G, M) = 0.$$

Sind also  $|G|$  und  $|M|$  teilerfremd, dann ist  $H^1(G, M) = 0$ .

**Beispiel 10** Sei  $G = S_3$ , in Magma also  $G := \text{SymmetricGroup}(3)$ . Mit ; und Enter eingeben.

Wir listen  $G$  auf:  $G\_list := [g : g \text{ in } G]$ . Die Liste können wir mittels `print G_list` betrachten.

```
[
  Id(G),
  (1, 2, 3),
  (1, 3, 2),
  (2, 3),
  (1, 2),
  (1, 3)
]
```

Wir können auch in  $G$  rechnen, e.g. gibt  $G!(1,2) * G!(2,3)$  das Ergebnis  $(1,3,2)$ . Magma komponiert also in der natürlichen Reihenfolge, nicht in der traditionellen, i.e. es wird zuerst  $G!(1,2)$  ausgeführt und dann  $G!(2,3)$ .

Sei  $M := \mathbb{Z}/2\mathbb{Z}$  der triviale  $G$ -Modul, in Magma also  $M := \text{Integers}(2)$ .

Wir bilden die Liste der Elemente von  $M$ :  $M\_list := [m : m \text{ in } M]$ . Mit `print M_list` erhalten wir folgendes.

```
[ 0, 1 ]
```

Wir bilden die Menge aller Abbildungen von  $G$  nach  $M$ , welche wir interpretieren als Tupel, in welchem der  $i$ -te Eintrag das Bild des  $i$ -ten Gruppenelements in  $G\_list$  sei.

$\text{Abb} := \text{CartesianPower}(M\_list, 6)$ .

Wir sortieren nun aus  $\text{Abb}$  die Derivationen aus.

Derivationen sind hier nun zugleich die Gruppenmorphisimen von  $G$  nach  $M$ , aber davon wollen wir keinen Gebrauch machen.

```
derivations := [];
for d in Abb do
  is_derivation := true;
  for i in [1..#G_list] do // i: Nummer des ersten zu testenden Gruppenelements
    for j in [1..#G_list] do // j: Nummer des zweiten zu testenden Gruppenelements
      if not d[i] + d[j] eq d[Index(G_list, G_list[i] * G_list[j])] then
        is_derivation := false;
        break i;
      end if;
    end for;
  end for;
  if is_derivation then
    derivations cat:= [d];
  end if;
end for;
print derivations;
```

Oder kürzer :

```
derivations :=  
[d : d in Abb | &and[d[i] + d[j] eq d[Index(G_list, G_list[i] * G_list[j])] : i,j in [1..#G_list]]];  
print derivations;
```

Beidemale erhalten wir [ <0, 0, 0, 0, 0, 0>, <0, 0, 0, 1, 1, 1> ].

Die einzige nichtverschwindende Derivation  $\delta$  bildet also die Elemente  $\text{Id}(G)$ ,  $(1, 2, 3)$ ,  $(1, 3, 2)$  auf 0 ab und die Elemente  $(2, 3)$ ,  $(1, 2)$ ,  $(1, 3)$  auf 1.

Wegen  $M$  trivial ist die Nullabbildung die einzige innere Derivation.

Also ist die Liste der Derivationen bereits die Liste der Elemente von  $H^1$ .

Folglich ist

$$H^1(G, M) \simeq \mathbb{Z}/2\mathbb{Z},$$

wobei bei der rechten Seite nur die additive Gruppe gemeint ist.

**Beispiel 11** Sei  $G := \text{SymmetricGroup}(3)$  und  $G\_list := [g : g \text{ in } G]$ .

Sei  $M := \text{Integers}(3)$  und  $M\_list := [m : m \text{ in } M]$  ein trivialer  $G$ -Modul.

Sei  $\text{Abb} := \text{CartesianPower}(M\_list, 6)$ . Wir sortieren nun aus  $\text{Abb}$  die Derivationen aus.

```
derivations :=  
[d : d in Abb | &and[d[i] + d[j] eq d[Index(G_list, G_list[i] * G_list[j])] : i,j in [1..6]]];  
print derivations;
```

Wir erhalten [ <0, 0, 0, 0, 0, 0> ], i.e.

$$H^1(G, M) \simeq 0.$$

**Beispiel 12** Sei  $G := \text{SymmetricGroup}(3)$  und  $G\_list := [g : g \text{ in } G]$ .

Sei  $M := \text{Integers}(3)$  und  $M\_list := [m : m \text{ in } M]$ .

Sei nun aber  $M$  kein trivialer  $G$ -Modul mehr, sei vielmehr

$$g \cdot m = \text{sgn}(g) \cdot m.$$

Sei  $\text{Abb} := \text{CartesianPower}(M\_list, 6)$ . Wir sortieren nun aus  $\text{Abb}$  die Derivationen aus.

```
derivations :=  
[d : d in Abb |  
&and[d[i] + Sign(G_list[i]) * d[j] eq d[Index(G_list, G_list[i] * G_list[j])] : i,j in [1..6]]];  
print derivations;
```

Wir erhalten

```
[ <0, 0, 0, 0, 0, 0>, <0, 0, 0, 1, 1, 1>, <0, 0, 0, 2, 2, 2>,
  <0, 1, 2, 0, 2, 1>, <0, 1, 2, 1, 0, 2>, <0, 1, 2, 2, 1, 0>,
  <0, 2, 1, 0, 1, 2>, <0, 2, 1, 1, 2, 0>, <0, 2, 1, 2, 0, 1> ] ,
```

i.e.

$$Z^1(G, M) \simeq \mathbb{Z}/3\mathbb{Z} \oplus \mathbb{Z}/3\mathbb{Z} .$$

Nun haben wir aber auch nichtverschwindende innere Derivationen.

```
inner_derivations :=
{<Sign(G_list[i]) * m - m : i in [1..6]> : m in M};
print inner_derivations;
```

Wir erhalten { <0, 0, 0, 0, 0, 0>, <0, 0, 0, 1, 1, 1>, <0, 0, 0, 2, 2, 2> },

i.e.

$$B^1(G, M) \simeq \mathbb{Z}/3\mathbb{Z} .$$

Folglich ist

$$H^1(G, M) = Z^1(G, M)/B^1(G, M) \simeq \mathbb{Z}/3\mathbb{Z} .$$

### 3 Umgang mit Gruppen

Mit `SmallGroups` erhält man eine Liste von Gruppen von vorgegebener Ordnung.

Seien e.g. `G := SmallGroups(6) [1]` und `H := SmallGroups(6) [2]` die beiden Gruppen von Ordnung 6.

Es hat `IsAbelian(G)` den Wert `false`, während `IsAbelian(H)` den Wert `true` hat. In der Tat hat auch `IsIsomorphic(G, SymmetricGroup(3))` den Wert `true`.

Mit `H_perm := PermutationGroup(FPGroup(H))` erhält man eine zu `H` isomorphe Untergruppe `H_perm` einer symmetrischen Gruppe. `Magma` konstruiert `H_perm` als Untergruppe der  $S_6$ , erzeugt von  $(1, 2)(3, 5)(4, 6)$  und  $(1, 3, 4)(2, 5, 6)$ . In der Tat gibt `[x : x in H_perm]` die Liste

```
[
  Id(H_perm),
  (1, 2)(3, 5)(4, 6),
  (1, 3, 4)(2, 5, 6),
  (1, 4, 3)(2, 6, 5),
  (1, 5, 4, 2, 3, 6),
  (1, 6, 3, 2, 4, 5)
] ,
```

was zeigt, daß `H_perm` die von  $(1, 5, 4, 2, 3, 6)$  erzeugte zyklische Gruppe ist.

$G_{\text{perm}} := \text{PermutationGroup}(\text{FPGroup}(G))$  liefert dagegen, wenig überraschend, die  $S_3$ , namentlich als die von  $(2,3)$  und  $(1,2,3)$  erzeugte Untergruppe von  $S_3$ .

Der Zwischenschritt  $G_{\text{fp}}\langle a, b \rangle := \text{FPGroup}(G)$  liefert die von  $a$  und  $b$  erzeugte Gruppe, modulo den Relationen  $a^2 = 1$ ,  $b^3 = 1$  und  $b^a = b^2$ . Mit anderen Worten,

$$G_{\text{fp}} = \langle a, b : a^2, b^3, b^a \cdot b \rangle$$

Hierbei ist  $b^a := a^{-1}ba$ .

Dies bedeutet, daß zum einen in  $G_{\text{fp}}$  die Gleichungen  $a^2 = 1$ ,  $b^3 = 1$  und  $b^a = b^2$  gelten. Zum anderen kann ein Gruppenmorphismus  $f$  von  $G_{\text{fp}}$  in eine andere Gruppe  $K$  dadurch eindeutig definiert werden, daß  $f(a), f(b) \in K$  so gewählt werden, daß in  $K$  die Gleichungen  $f(a)^2 = 1$ ,  $f(b)^3 = 1$  und  $f(b)^{f(a)} = f(b)^2$  gelten.

Z.B. kann ein Gruppenmorphismus  $f$  von  $G_{\text{fp}}$  nach  $G_{\text{perm}}$  durch  $f(a) := (2,3)$  und  $f(b) := (1,2,3)$  definiert werden.

Mehr Details hierzu in §4.

Es ist  $G_{\text{fp}}$  zu  $G_{\text{perm}}$  isomorph. Diesen Isomorphismus wird von Magma geliefert, wenn wir uns  $G_{\text{perm}}$  nochmals definieren, und dabei als zweite Ausgabe einen Isomorphismus  $\text{phi}$  mitliefern lassen:  $G_{\text{perm}}, \text{phi} := \text{PermutationGroup}(G_{\text{fp}})$ . Es ist dann  $\text{phi}$  ein Gruppenisomorphismus von  $G_{\text{fp}}$  nach  $G_{\text{perm}}$ , für welchen  $\text{phi}(a)$  zu  $(2,3)$  und  $\text{phi}(b)$  zu  $(1,2,3)$  wird.

Sowieso wird  $\text{phi}(G_{\text{fp}}!1)$  zu  $\text{Id}(G_{\text{perm}})$ . Es wird aber z.B. auch  $\text{phi}(a*b)$  zu  $(1,2)$ .

Rückwärts wird  $(G_{\text{perm}}!(1,2))@@\text{phi}$  wieder zu  $a*b$ .

## 4 Gruppen in Erzeugern und Relationen

Wir wollen eine Gruppen mit vorgegebenen Erzeugern und vorgegebenen Relationen definieren. Der erste Schritt ist dabei die Konstruktion der freien Gruppe.

**Definition 13** Sei  $n \geq 1$ . Seien  $x_1, x_2, \dots, x_n$  paarweise verschiedene Elemente.

Wir bilden die Elemente  $x_1^{-1}, x_2^{-1}, \dots, x_n^{-1}$ , was zunächst nur eine Schreibweise sei.

Zusammen haben wir so eine Menge

$$A := \{x_1, x_2, \dots, x_n, x_1^{-1}, x_2^{-1}, \dots, x_n^{-1}\}$$

aus  $2n$  Elementen.

Sei  $F_0$  die Menge der Wörter in  $A$ .

Es trägt  $F_0$  als Multiplikation die Operation des Aneinandersetzens von Wörtern.

Das leere Wort wird  $1_F$  oder kurz  $1$  geschrieben.

Sei  $(\approx)$  die Relation auf  $F_0$ , die aus den Paaren der Form

$$(wx_k x_k^{-1} w', ww')$$

und den Paaren der Form

$$(wx_k^{-1} x_k w', ww')$$

besteht, wobei  $w, w' \in F_0$  und wobei  $k \in [1, n]$ .

Sei  $(\sim)$  die von  $(\approx)$  erzeugte Äquivalenzrelation.

Sei  $F := F_0/(\sim)$  die Menge der Äquivalenzklassen.

Es wird  $F$  eine Gruppe, indem man die Multiplikation repräsentantenweise durch Multiplikation in  $F_0$  definiert.

Es heißt  $F$  die *freie Gruppe* auf  $x_1, x_2, \dots, x_n$ .

Die Elemente von  $F$  werden repräsentantenweise notiert, i.e. ohne Kennzeichnung der Bildung der Äquivalenzklassen.

**Beispiel 14** Sei  $n := 2$ . Sei  $F$  die freie Gruppe auf  $x_1$  und  $x_2$ .

In  $F$  wird

$$(x_1 x_2 x_2 x_1^{-1}) \cdot (x_1 x_2^{-1} x_1 x_1 x_1) = x_1 x_2 x_1 x_1 x_1.$$

Weitere Kürzungen sind nicht mehr möglich. Man kann nur noch zusammenfassen zu

$$x_1 x_2 x_1 x_1 x_1 = x_1 x_2 x_1^3.$$

**Bemerkung 15** Sei  $n \geq 1$ . Sei  $F$  die freie Gruppe auf  $x_1, x_2, \dots, x_n$ .

Sei  $T$  eine Gruppe. Seien  $t_1, t_2, \dots, t_n \in T$ .

Es gibt genau einen Gruppenmorphismus  $\varphi : F \rightarrow T$  mit  $\varphi(x_k) = t_k$  für  $k \in [1, n]$ .

**Definition 16** Sei  $n \geq 1$ . Sei  $F$  die freie Gruppe auf  $x_1, x_2, \dots, x_n$ .

Sei  $r \geq 1$ . Seien  $w_1, \dots, w_r \in F$ .

Sei  $N$  die Untergruppe von  $F$ , die von der Menge  $\{w_k^v : k \in [1, r], v \in F\}$  erzeugt wird. Es ist  $N \trianglelefteq F$ .

Sei

$$G := \langle x_1, x_2, \dots, x_n : w_1, \dots, w_r \rangle := F/N = \{vN : v \in F\}$$

die Gruppe, die von den *Erzeugern*  $x_1, x_2, \dots, x_n$  modulo den *Relationen*  $w_1, \dots, w_r$  definiert ist.

Es sei  $\rho : F \rightarrow G : v \mapsto vN$  der Restklassenmorphismus.

Häufig wird statt  $vN$  einfach nur  $v$  geschrieben, falls aus dem Kontext hervorgeht, daß ein Element in  $G$  gemeint ist.

**Bemerkung 17** Sei  $n \geq 1$ . Sei  $F$  die freie Gruppe auf  $x_1, x_2, \dots, x_n$ .

Sei  $r \geq 1$ . Seien  $w_1, \dots, w_r \in F$ . Sei

$$G := \langle x_1, x_2, \dots, x_n : w_1, \dots, w_r \rangle.$$

Sei  $T$  eine Gruppe. Seien  $t_1, t_2, \dots, t_n \in T$ .

Es gibt genau einen Gruppenmorphismus  $\varphi : F \rightarrow T$  mit  $\varphi(x_k) = t_k$  für  $k \in [1, n]$ ; cf. Bemerkung 15.

Sei nun  $\varphi(w_k) = 1$  für  $k \in [1, r]$ .

Es gibt genau einen Gruppenmorphismus  $\psi : G \rightarrow T$  mit  $\psi(\rho(x_k)) = t_k$  für  $k \in [1, n]$ .

**Beispiel 18** Sei  $F\langle x_1, x_2 \rangle := \text{FreeGroup}(2)$ .

Sei  $G\langle y_1, y_2 \rangle, \rho := \text{quo}\langle F \mid x_1^5, x_2^3, (x_1 * x_2)^2 \rangle$ . Es ist  $\text{Order}(G)$  gleich 60.

Es ist  $\rho$  der Restklassenmorphismus von  $F$  nach  $G$ , und für diesen ist nun  $\rho(x_1)$  gleich  $y_1$ , und  $\rho(x_2)$  gleich  $y_2$ .

In  $G$  kann er aber noch nicht vernünftig rechnen. E.g. erkennt er  $y_1^5$  nicht als  $1_G$ , obwohl dies richtig ist.

Sei daher noch  $G_{\text{perm}}, \sigma := \text{PermutationGroup}(G)$ . Dann ist  $G_{\text{perm}}$  eine Untergruppe der  $S_6$  und  $\sigma$  ein Isomorphismus von  $G$  nach  $G_{\text{perm}}$ . Wir setzen  $z_1 := \sigma(y_1)$  und  $z_2 := \sigma(y_2)$ . Dann verschwinden nun in der Tat  $z_1^5, z_2^3$  und  $(z_1 * z_2)^2$ .

Magma hat nun zur Konstruktion von  $\sigma$  Gebrauch von Bemerkung 17 gemacht: Es wurde von Magma  $z_1$  zu  $(2, 3, 4, 5, 6)$  und  $z_2$  zu  $(1, 2, 3) (4, 6, 5)$  gewählt, wobei Magma darauf geachtet hat, daß  $z_1^5, z_2^3$  und  $(z_1 * z_2)^2$  verschwinden. Dank Bemerkung 17 existiert ein Gruppenmorphismus von  $G$  nach  $S_6$ , der  $y_1$  nach  $z_1$  und der  $y_2$  nach  $z_2$  schickt. Schränkt man diesen im Zielbereich auf das Erzeugnis  $G_{\text{perm}}$  von  $z_1$  und  $z_2$  ein, so entsteht der surjektive Gruppenmorphismus  $\sigma$  von  $G$  nach  $G_{\text{perm}}$ . Magma hat zudem darauf geachtet, daß  $\sigma$  auch injektiv ist, insgesamt also ein Isomorphismus.

## 5 Berechnung von $H^1$ für Gruppen in Erzeugern und Relationen

Sei  $n \geq 1$ . Sei  $F$  die freie Gruppe auf  $x_1, x_2, \dots, x_n$ .

Sei  $r \geq 1$ . Seien  $w_1, \dots, w_r \in F$ .

Sei

$$G := \langle x_1, x_2, \dots, x_n : w_1, \dots, w_r \rangle.$$

Wir haben den Gruppenmorphimus

$$\begin{aligned} F &\xrightarrow{\rho} G \\ x_i &\mapsto x_i \quad \text{für } i \in [1, n] \end{aligned}$$

Sei  $M$  ein  $G$ -Modul, der als  $\mathbb{Z}$ -Modul endlich erzeugt frei von Rang  $\text{rk } M$  sei. Wenn nötig, wird als abelsche Gruppe  $M = \mathbb{Z}^{(\text{rk } M) \times 1}$  identifiziert.

Via  $y \cdot m := \rho(y) \cdot m$  für  $y \in F$  und  $m \in M$  ist  $M$  auch ein  $F$ -Modul.

Insbesondere wird  $z \cdot m = \rho(z) \cdot m = 1 \cdot m = m$  für  $z \in \text{Kern}(\rho)$  und  $m \in M$ .

**Bemerkung 19** Seien  $m_1, m_2, \dots, m_n \in M$  gegeben.

Es gibt genau eine Derivation  $\delta \in Z^1(F, M)$  mit  $\delta(x_j) = m_j$  für  $j \in [1, n]$ .

Diese erfüllt

$$\delta(x_j^{-1}) = -x_j^{-1} \cdot \delta(x_j)$$

und

$$\delta(x_{i_1}^{\varepsilon_1} \cdot \dots \cdot x_{i_t}^{\varepsilon_t}) = \sum_{s \in [1, t]} x_{i_1}^{\varepsilon_1} \cdot \dots \cdot x_{i_{s-1}}^{\varepsilon_{s-1}} \cdot m_{i_s, \varepsilon_s}$$

für  $t \geq 0$ , für  $i_1, \dots, i_t \in [1, n]$  und für  $\varepsilon_1, \dots, \varepsilon_t \in \{-1, +1\}$ .

Dann definieren wir  $m_{j,1} := m_j$  und  $m_{j,-1} := -x_j^{-1} \cdot m_j$  für  $j \in [1, n]$  und eine Abbildung  $\delta_0 : F_0 \rightarrow M$  via

$$\delta_0(x_{i_1}^{\varepsilon_1} \cdot \dots \cdot x_{i_t}^{\varepsilon_t}) := \sum_{s \in [1, t]} x_{i_1}^{\varepsilon_1} \cdot \dots \cdot x_{i_{s-1}}^{\varepsilon_{s-1}} \cdot m_{i_s, \varepsilon_s},$$

so wird

$$\delta_0(w \cdot x_j \cdot x_j^{-1} \cdot w') = \delta_0(w \cdot w')$$

und

$$\delta_0(w \cdot x_j^{-1} \cdot x_j \cdot w') = \delta_0(w \cdot w')$$

für  $w, w' \in F_0$  und  $j \in [1, n]$ .

Also ist  $\delta : F \rightarrow M$  durch  $\delta(w) := \delta_0(w)$  für  $w \in F_0$  definierbar, wobei das Argument von  $\delta$  vereinbarungsgemäß bereits die Äquivalenzklasse von  $w$  in  $F$  bezeichne.

Die definierende Formel gibt dann auch

$$\delta(w \cdot w') = \delta(w) + w \cdot \delta(w').$$

Also ist  $\delta \in Z^1(F, M)$ .

Die Eindeutigkeit von  $\delta$  folgt aus der Tatsache, daß die angegebene Formel für  $\delta$  notwendigerweise gilt.

## Beispiel 20 Wir nehmen

```
G := SymmetricGroup(3);
G_fp<a,b>,phi := FPGroup(G);
print phi(a), phi(b); // gibt (1,2,3) und (1,2)
F<A,B> := FreeGroup(2);
rho := hom< F -> G_fp | [<A,a>,<B,b>] >; // Restklassenmorphismus
n := #Generators(G_fp);
```

Wir beschaffen uns  $G$ -Moduln wie folgt.

```
IM_rat := IrreducibleModules(G,Rationals());
M_rat := IM_rat[3];
rk := Dimension(M_rat);
MR := MatrixRing(Integers(),rk);
```

Es operiert  $\phi(a)$  via der Matrix  $\text{ActionGenerator}(M\_rat,1)$ . Es operiert  $\phi(b)$  via der Matrix  $\text{ActionGenerator}(M\_rat,2)$ .

Wir bilden eine Liste dieser Matrizen.

```
rep_mat := [MatrixRing(Integers(),Dimension(M_rat))!ActionGenerator(M_rat,i) :
            i in [1..#Generators(G)]];
```

Damit bauen wir einen  $G$ -Modul mit Grundring  $\mathbb{Z}$ .

```
M := GModule(G,rep_mat);
```

Es ist dann  $M = \mathbb{Z}^{2 \times 1}$ , worauf die gewählten Erzeuger von  $G$  mit den Matrizen aus  $\text{rep\_mat}$  operieren.

Wir bauen uns die Multiplikation eines Elements  $w \in F$  auf  $M$ .

```
prod := function(w,m)
  // w: Element von F
  // m: Element von M
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  return &*([MR!1] cat [rep_mat[index_seq[i]]^sign_seq[i] : i in [1..t]]) * m;
end function;
```

Wir testen das:

```
prod(A,Matrix([[1],[0]]));
/* gibt
[-1]
[-1]
*/
prod(A,Matrix([[0],[1]]));
/* gibt
[1]
[0]
*/
```

Das paßt auch zur Matrix `rep_mat[1]`;

Noch ein Test.

```
w1 := B * A^-2 * B * A^-1;
w2 := A * B^-1 * A^3;
prod(w1,prod(w2,Matrix([[3],[5]]))) eq prod(w1 * w2,Matrix([[3],[5]])); // gibt true
```

Folgendermaßen erhalten wir für eine Matrix  $U \in \mathbb{Z}^{(\text{rk } M) \times n} = \mathbb{Z}^{2 \times 2}$  eine Derivation  $\delta \in Z^1(F, M)$ , die den ersten Erzeuger  $A$  auf die erste Spalte von  $U$  schickt und die den zweiten Erzeuger  $b$  auf die zweite Spalte von  $U$  schickt, jeweils gesehen als Elemente von  $M$ .

Sei noch `umnum := map<{-1,+1} -> {1,2} | [<1,1>, <-1,2>]>` global festgelegt.

Sei nun die angekündigte Derivation wie folgt konstruiert.

```
derivation := function(U,w)
  // U: ganzzahlige Matrix, Spalten in M, Spaltenzahl gleich Erzeugerzahl
  // w: abzubildendes Element von F
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  elt_seq := [[ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]],
    [-rep_mat[index_seq[i]]^-1 * ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]]];
  return &+([RMatrixSpace(Integers(),rk,1)!0] cat
    [&*(MR!1] cat [rep_mat[index_seq[j]]^sign_seq[j] : j in [1..i-1]])
    * elt_seq[umnum(sign_seq[i])][i] : i in [1..#w_seq]]);
end function;
```

Wir testen das:

```
U := Matrix([[1,2],[3,4]]);
w1 := B * A^-2 * B * A^-1;
w2 := A * B^-1 * A^3;
w3 := A^-1 * B^-1 * A^3 * B;
```

```

derivation(U,w1 * w2) eq derivation(U,w1) + prod(w1,derivation(U,w2)); // gibt true
derivation(U,w1 * w2 * w3) eq derivation(U,w1)
    + prod(w1,derivation(U,w2))
    + prod(w1 * w2,derivation(U,w3)); // gibt true

```

**Bemerkung 21** Seien  $m_1, m_2, \dots, m_n \in M$  gegeben.

Sei  $\delta \in Z^1(F, M)$  mit  $\delta(x_j) = m_j$  für  $j \in [1, n]$ .

Es gibt genau dann eine Derivation  $\bar{\delta} : G \rightarrow M$  mit  $\bar{\delta}(\rho(x_j)) = m_j$  für  $j \in [1, n]$ , wenn  $\delta(w_i) = 0$  ist für  $i \in [1, n]$ .

Dazu setzen wir  $\bar{\delta}(\rho(y)) := \delta(y)$  für  $y \in F$ .

Zu zeigen ist nur, daß  $\bar{\delta}$  wohldefiniert ist, denn dann vererbt sich die Derivationseigenschaft von  $\delta$  auf  $\bar{\delta}$ .

Seien also  $y, \tilde{y} \in F$  mit  $\rho(y) = \rho(\tilde{y})$  gegeben. Zu zeigen ist  $\delta(y) \stackrel{!}{=} \delta(\tilde{y})$ .

Sei  $N \trianglelefteq F$  definiert wie in Definition 16. Es ist dann  $N = \text{Kern}(\rho)$ .

Es ist  $\rho(y^{-1} \cdot \tilde{y}) = \rho(y)^{-1} \cdot \rho(\tilde{y})$ , also  $z := y^{-1} \cdot \tilde{y} \in N$ . Es ist  $\tilde{y} = y \cdot z$ . Also ist

$$\delta(\tilde{y}) = \delta(y \cdot z) = \delta(y) + y \cdot \delta(z).$$

Es genügt also,  $\delta(z) \stackrel{!}{=} 0$  zu zeigen.

Dazu genügt es,  $\delta(w_k^v) \stackrel{!}{=} 0$  zu zeigen für  $k \in [1, r]$  und  $v \in F$ .

Es wird

$$\begin{aligned}
\delta(w_k^v) &= \delta(v^{-1} \cdot w_k \cdot v) \\
&= \delta(v^{-1}) + v^{-1} \cdot \delta(w_k) + v^{-1} \cdot w_k \cdot \delta(v) \\
&= -v^{-1} \cdot \delta(v) + v^{-1} \cdot w_k \cdot \delta(v) \\
&= -v^{-1} \cdot \delta(v) + v^{-1} \cdot \delta(v) \\
&= 0.
\end{aligned}$$

**Bemerkung 22 (Skizze des Vorgehens)**

Um Bemerkung 21 umzusetzen, müssen wir testen, welche Matrizen  $U \in \mathbb{Z}^{(\text{rk } M) \times n}$  eine Derivation  $\delta_U$  liefern, die auf allen Relationen  $w_1, \dots, w_r$  verschwindet.

Dazu benötigen wir etwas Lineare Algebra über  $\mathbb{Z}$ .

Wir schreiben  $U$  als Spalte  $\text{Col}_U \in \mathbb{Z}^{(\text{rk } M) \cdot n \times 1}$ , indem wir  $U$  zeilenweise auslesen und in die Spalte  $\text{Col}_U$  einlesen.

Wir werden uns in den unten folgenden Beispielen eine Matrix

$$Z \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot n}$$

derart, daß  $Z \cdot \text{Col}_U$  genau dann verschwindet, wenn  $\delta_U$  eine Derivation von  $G$  in  $M$  liefert.

Wir werden also  $Z^1(G, M)$  isomorph durch den Kern der  $\mathbb{Z}$ -linearen Abbildung

$$\begin{array}{ccc} \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{Z \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot r \times 1} \\ \text{Col}_U & \mapsto & Z \cdot \text{Col}_U \end{array}$$

ersetzen.

Wir werden ferner eine Matrix

$$I \in \mathbb{Z}^{(\text{rk } M) \cdot n \times (\text{rk } M)}$$

bauen, die einem Element  $m \in M = \mathbb{Z}^{(\text{rk } M) \times 1}$  die zu  $m \in M$  gehörige innere Derivation  $\delta_U$  zuweist, wobei sich für diese Matrix  $U$

$$\text{Col}_U = I \cdot m$$

ergibt.

Wir werden also  $B^1(G, M)$  isomorph durch das Bild der  $\mathbb{Z}$ -linearen Abbildung

$$\begin{array}{ccc} \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} \\ m & \mapsto & I \cdot m \end{array}$$

ersetzen.

Da jede innere Derivation eine Derivation ist, ist  $Z \cdot I = 0$ .

Der Elementarteilersatz liefert dann invertierbare Matrizen  $S \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot r}$  und  $T \in \mathbb{Z}^{(\text{rk } M) \cdot n \times (\text{rk } M) \cdot n}$  derart, daß

$$D := S \cdot Z \cdot T \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot n}$$

eine Diagonalmatrix ist, i.e. an allen Positionen mit Zeilenindex ungleich Spaltenindex einen Nulleintrag hat.

Sei

$$I' := T^{-1} \cdot I$$

So werden wir  $Z^1(G, M)$  isomorph durch den Kern von  $D \cdot (-)$  und  $B^1(G, M)$  durch das Bild von  $I' \cdot (-)$  ersetzt haben.

$$\begin{array}{ccccc} \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{Z \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot r \times 1} \\ \parallel & & \uparrow \wr T \cdot (-) & & \downarrow \wr S \cdot (-) \\ \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I' \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{D \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot r \times 1} \end{array}$$

Der Kern von  $D \cdot (-)$  besteht aus den Elementen von  $\mathbb{Z}^{(\text{rk } M) \cdot n \times 1}$ , die in den Positionen 1 bis  $\text{rk } D$  einen Nulleintrag haben. Wir schreiben

$$c := (\text{rk } M) \cdot n - (\text{rk } D)$$

und definieren dementsprechend folgende Blockmatrix.

$$J := \begin{pmatrix} 0 \\ E_c \end{pmatrix} \in \mathbb{Z}^{(\text{rk } M) \cdot n \times c}.$$

Dann ist das Bild von  $J \cdot (-)$  gleich dem Kern von  $D \cdot (-)$ .

Es ist nun  $D \cdot I' = 0$ . Ist also

$$I'' \in \mathbb{Z}^{c \times (\text{rk } M)}$$

die Matrix, die aus  $I'$  durch Weglassen der Nullzeilen in Position 1 bis  $\text{rk } D$  entsteht, so ist

$$J \cdot I'' = I'.$$

Nun ist  $Z^1(G, M)$  isomorph ersetzt durch  $\mathbb{Z}^{c \times 1}$  und  $B^1(G, M)$  durch das Bild von  $I'' \cdot (-)$ .

Der Elementarteilersatz liefert dann invertierbare Matrizen  $S'' \in \mathbb{Z}^{c \times c}$  und  $T'' \in \mathbb{Z}^{(\text{rk } M) \times (\text{rk } M)}$  derart, daß

$$D'' := S'' \cdot I'' \cdot T'' \in \mathbb{Z}^{c \times (\text{rk } M)}$$

eine Diagonalmatrix ist.

$$\begin{array}{ccccc}
 \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{Z \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot r \times 1} \\
 \parallel & & \uparrow \wr T \cdot (-) & & \downarrow \wr S \cdot (-) \\
 \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I' \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{D \cdot (-)} & \mathbb{Z}^{(\text{rk } M) \cdot r \times 1} \\
 \parallel & & \uparrow J \cdot (-) & & \\
 \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{I'' \cdot (-)} & \mathbb{Z}^{c \times 1} & & \\
 \uparrow \wr T'' \cdot (-) & & \downarrow \wr S'' \cdot (-) & & \\
 \mathbb{Z}^{(\text{rk } M) \times 1} & \xrightarrow{D'' \cdot (-)} & \mathbb{Z}^{c \times 1} & & 
 \end{array}$$

Nun ist  $Z^1(G, M)$  isomorph ersetzt durch  $\mathbb{Z}^{c \times 1}$  und  $B^1(G, M)$  durch das Bild von  $D'' \cdot (-)$ .

Mit anderen Worten, sind  $d''_1, \dots, d''_c \in \mathbb{Z}$  die Diagonaleinträge von  $D''$ , dann ist

$$H^1(G, M) \simeq \mathbb{Z}/d''_1\mathbb{Z} \oplus \dots \oplus \mathbb{Z}/d''_c\mathbb{Z}.$$

Aus der Theorie weiß man auch, daß  $\text{rk } I + \text{rk}(\text{Kern}(Z \cdot (-))) = (\text{rk } M) \cdot n$  ist. Wir hätten also auch direkt die Elementarteiler von  $I$  anstelle der Elementarteiler  $d''_1, \dots, d''_c$  von  $I''$  heranziehen können. Wir wollen hier aber von dieser Theorieaussage keinen Gebrauch machen, sondern sie wie oben geschildert im jeweiligen Beispiel verifizieren: es sollte  $\text{rk } D'' = c$  sein, i.e. es sollten alle Elemente  $d''_1, \dots, d''_c$  ungleich 0 sein.

**Beispiel 23** Wir setzen Beispiel 20 fort.

Es hängt  $\delta_U(w) := \text{derivation}(U, w)$  linear von der Matrix  $U$  ab, welche  $rk$  Zeilen und  $n$  Spalten hat.

Wir müssen ermitteln, welche Matrizen  $U$  eine Derivation  $\delta_U$  liefern, die  $\delta_U(w_k) = 0$  gibt für  $k \in [1, n]$ .

Wir besorgen uns die Relationen unserer Gruppe  $G_{fp}$ .

```
w := [x[1] : x in Relations(G_fp)];  
r := #w;
```

Es ist also  $w_k = w[k]$  für  $k \in [1, r]$ .

Wir ersetzen  $U$  durch einen Spaltenvektor  $Col\_U$ , für welchen dann im nachhinein wieder  $U[i, j] := Col\_U[(i-1) * n + j]$  sein soll.

Wir produzieren uns Matrizen mit nur einer 1 als Eintrag, auch Elementarmatrizen genannt.

```
U_elem := function(i, j, rk)  
  U := RMatrixSpace(Integers(), rk, n)!0;  
  U[i, j] := 1;  
  return U;  
end function;
```

Es ist nun  $\text{derivation}(U, w[k])$  gleich der Summe der Elemente

$$\text{derivation}(U\_elem(i, j, rk), w[k]) * U[i, j] ,$$

i.e. der Elemente

$$\text{derivation}(U\_elem(i, j, rk), w[k]) * Col\_U[(i-1) * n + j] .$$

Wir bauen nun eine Matrix  $Z$  derart, daß das Produkt  $Z * Col\_U$  genau dann verschwindet, wenn  $\text{derivation}(U, w[k])$  verschwindet für alle  $k \in [1, r]$ .

```
Z := RMatrixSpace(Integers(), rk*r, rk*n)!0;  
for k in [1..r] do  
  for i in [1..rk] do  
    for j in [1..n] do  
      InsertBlock(~Z, derivation(U_elem(i, j, rk), w[k]), 1 + (k-1)*rk, (i-1) * n + j);  
    end for;  
  end for;  
end for;
```

```

print Z;
/* gibt
[ 0  1  0  1]
[ 0  1  0  1]
[ 0  0  0  0]
[ 0  0  0  0]
[ 0  0  0  0]
[-2  2  1 -1]
*/

```

Nun noch die inneren Derivationen.

```

U_inn := function(m) // m in M
  U := RMatrixSpace(Integers(),rk,n)!0;
  for j in [1..n] do
    InsertBlock(~U, rep_mat[j] * m - m, 1, j);
  end for;
  return U;
end function;

```

Wir lassen  $m$  über eine Basis von  $M$  laufen und schreiben die resultierenden Matrizen  $U$  als Spalten in eine Matrix  $I$ .

```

I := RMatrixSpace(Integers(),rk*n,rk)!0;
for k in [1..rk] do
  m := RMatrixSpace(Integers(),rk,1)!0;
  m[k,1] := 1;
  U := U_inn(m);
  for i in [1..rk] do
    for j in [1..n] do
      I[(i-1) * n + j,k] := U[i,j];
    end for;
  end for;
end for;
print I;
/*
[-2  1]
[-1  1]
[-1 -1]
[ 1 -1]
*/

```

Innere Derivationen sollten Derivationen sein: in der Tat gibt  $Z * I$  die Nullmatrix.

Sei nun

```

D,S,T := SmithForm(Z);

```

Dann sind  $S$  und  $T$  invertierbare ganzzahlige Matrizen mit  $S * Z * T = D$  diagonal. Bei uns:

```

> D;
[1 0 0 0]
[0 1 0 0]
[0 0 0 0]
[0 0 0 0]
[0 0 0 0]
[0 0 0 0]
> S;
[ 1  0  0  0  0  0]
[-2  0  0  0  0  1]
[ 0  0  1  0  0  0]
[ 0  0  0  1  0  0]
[ 0  0  0  0  1  0]
[-1  1  0  0  0  0]
> T;
[ 0  0  1  0]
[ 1  0  0 -1]
[ 0  1  2  3]
[ 0  0  0  1]

```

Es ist genau dann  $Zx = 0$ , wenn  $D(T^{-1}x) = SZTT^{-1}x = 0$  ist. Dies ist genau dann der Fall, wenn  $T^{-1}x$  mit  $\text{rkD} := \text{Rank}(D)$  Nulleinträgen beginnt.

So etwa ist  $T^{-1} * I$  eine Matrix mit Nullen in den den ersten beiden Zeilen.

Sei  $II := \text{RowSubmatrixRange}(T^{-1} * I, \text{rkD} + 1, \text{rk} * n)$ .

Sei

```

DD,SS,TT := SmithForm(II);
print DD;
/*
[1 0]
[0 1]
*/

```

Folglich ist  $B^1(G, M) = Z^1(G, M)$  und also

$$H^1(G, M) = 0.$$

Wir vergleichen das Resultat noch kurz mit dem des in Magma eingebauten Befehls. Vorsicht, Magma arbeitet mit Rechtsmoduln, daher muß zwecks Vergleich dualisiert werden.

```

rep_mat_dual := [Transpose(x)^-1 : x in rep_mat];
CM := CohomologyModule(G, GModule(G, rep_mat_dual));
CohomologyGroup(CM, 1);

```

```

/*
Full Quotient RSpace of degree 0 over Integer Ring
Column moduli:
[ ]
*/

```

**Beispiel 24** Wir variieren Beispiel 20, 23.

Der Vorspann war folgender.

```

G := SymmetricGroup(3);
G_fp<a,b>,phi := FPGroup(G);
print phi(a), phi(b); // gibt (1,2,3) und (1,2)
F<A,B> := FreeGroup(2);
rho := hom< F -> G_fp | [<A,a>,<B,b>] >; // Restklassenmorphismus
n := #Generators(G_fp);
IM_rat := IrreducibleModules(G,Rationals());
M_rat := IM_rat[3];
rk := Dimension(M_rat);
MR := MatrixRing(Integers(),rk);
rep_mat := [MR!ActionGenerator(M_rat,i) : i in [1..#Generators(G)]];
M := GModule(G,rep_mat);

prod := function(w,m)
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  return &*([MR!1] cat [rep_mat[index_seq[i]]^sign_seq[i] : i in [1..t]]) * m;
end function;

umnum := map<{-1,+1} -> {1,2} | [<1,1>, <-1,2>]>;
w := [x[1] : x in Relations(G_fp)];
r := #w;

U_elem := function(i,j,rk)
  U := RMatrixSpace(Integers(),rk,n)!0;
  U[i,j] := 1;
  return U;
end function;

```

Wir ersetzen nun  $M$  durch den zu  $M$  dualen Modul  $M^*$ . Dazu müssen wir ersetzen:

```

rep_mat_M := rep_mat;
rep_mat_M_dual := [Transpose(x)^-1 : x in rep_mat_M];
rep_mat := rep_mat_M_dual;

```

Wir wiederholen:

```
derivation := function(U,w)
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  elt_seq := [[ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]],
    [-rep_mat[index_seq[i]]^-1 * ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]]];
  return &+([RMatrixSpace(Integers(),rk,1)!0] cat
    [&*( [MR!1] cat [rep_mat[index_seq[j]]^sign_seq[j] : j in [1..i-1]]
    * elt_seq[umnum(sign_seq[i])] [i] : i in [1..#w_seq]]]);
end function;
```

```
Z := RMatrixSpace(Integers(),rk*r,rk*n)!0;
for k in [1..r] do
  for i in [1..rk] do
    for j in [1..n] do
      InsertBlock(~Z,derivation(U_elem(i,j,rk),w[k]), 1 + (k-1)*rk, (i-1) * n + j);
    end for;
  end for;
end for;
print Z;
/* gibt
[ 0  1  0  1]
[ 0  1  0  1]
[ 0  0  0  0]
[ 0  0  0  0]
[ 1 -1  0  0]
[-2  2  0  0]
*/
```

```
U_inn := function(m) // m in M
  U := RMatrixSpace(Integers(),rk,n)!0;
  for j in [1..n] do
    InsertBlock(~U, rep_mat[j] * m - m, 1, j);
  end for;
  return U;
end function;
```

```
I := RMatrixSpace(Integers(),rk*n,rk)!0;
for k in [1..rk] do
  m := RMatrixSpace(Integers(),rk,1)!0;
  m[k,1] := 1;
  U := U_inn(m);
  for i in [1..rk] do
    for j in [1..n] do
      I[(i-1) * n + j,k] := U[i,j];
    end for;
  end for;
end for;
```

```

print I;
/*
[-1  1]
[-1  1]
[-1 -2]
[ 1 -1]
*/

D,S,T := SmithForm(Z);
rkD := Rank(D);
II := RowSubmatrixRange(T^-1 * I, rkD + 1, rk * n);
DD,SS,TT := SmithForm(II);
print DD;
/*
[1 0]
[0 3]
*/

```

Also ist  $H^1(G, M^*) \simeq \mathbb{Z}/3\mathbb{Z}$ .

Sei noch

```

J := RMatrixSpace(Integers(), rk*n, rk*n - rkD)!0;
InsertBlock(~J, MatrixRing(Integers(), rk*n - rkD)!1, rkD+1, 1);

```

die Einbettungsmatrix. Dann ist  $J * II \text{ eq } T^{-1} * I$ .

Wir wollen nun noch einen Erzeuger dieser Gruppe von einer Derivation repräsentieren lassen.

Dem Element  $\begin{pmatrix} 0 \\ 1 \end{pmatrix}$ , also  $e2 := \text{Matrix}(\begin{bmatrix} 0 \\ 1 \end{bmatrix})$ , entspricht das Element  $T * J * SS^{-1} * e2$ , also

```

/*
[ 0]
[-1]
[ 3]
[ 1]
*/ .

```

Übersetzt wird diese Spalte zur folgenden Matrix.

```

U_rep := Matrix(\begin{bmatrix} 0 & -1 \\ 3 & 1 \end{bmatrix})

```

Wir bauen daraus eine Derivation auf  $G$ .

```

d_rep := function(U_rep,g);
  return derivation(U_rep,(G!g)@@phi@@rho);
end function;

```

Wir testen, ob `d_rep` mittels `U_rep` tatsächlich eine Derivation auf  $G$  liefert.

```
M_dual := GModule(G,rep_mat);
rep := Representation(M_dual);
&and[d_rep(U_rep,g) + rep(g) * d_rep(U_rep,h) eq d_rep(U_rep,g*h) : g,h in G];
/*
true
*/
print [<g, d_rep(U_rep,g)> : g in G];
// alle Gruppenelemente samt Bildern
/*
[
  <
    Id(G),
    [0]
    [0]
  >,
  <
    (1, 2, 3),
    [0]
    [3]
  >,
  <
    (1, 3, 2),
    [3]
    [3]
  >,
  <
    (2, 3),
    [2]
    [4]
  >,
  <
    (1, 2),
    [-1]
    [ 1]
  >,
  <
    (1, 3),
    [2]
    [1]
  >
]
```

```
]
*/
```

**Beispiel 25** Wir variieren Beispiel 24 und wollen die erste Cohomologiegruppe des regulären Moduls  $\mathbb{Z}G$  bestimmen; cf. Beispiel 3.

Es ist aus der Theorie bekannt, daß  $H^1(G, \mathbb{Z}G) = 0$  ist. Wir wollen hier also die Übereinstimmung von Theorie und Praxis prüfen.

Für eine Permutationsgruppe  $G$  mit festgelegter Auflistung `G_list` erhalten wir die beschreibende Matrix eines Elements  $x \in G$  wie folgt.

```
rep_mat_reg := function(G,G_list,x)
// G: Permutationsgruppe
// x: Element von G
n := Order(G);
A := MatrixRing(Integers(),n)!0;
for i in [1..n] do
  A[Index(G_list, x * G_list[i]), i] := 1;
end for;
return A;
end function;
```

Wie in Beispiel 24 erhalten wir nun nach Eingabe des dortigen Vorspanns folgendes.

```
G_list := [x : x in G];
rep_mat := [rep_mat_reg(G,G_list,phi(a)),rep_mat_reg(G,G_list,phi(b))];
M_reg := GModule(G,rep_mat);
rk := Order(G);
MR := MatrixRing(Integers(),rk);
```

Nach weiterem Vorgehen wie in Beispiel 24 erhalten wir folgendes.

```
DD;
/*
[1 0 0 0 0 0]
[0 1 0 0 0 0]
[0 0 1 0 0 0]
[0 0 0 1 0 0]
[0 0 0 0 1 0]
*/
```

Also ist in der Tat  $H^1(G, \mathbb{Z}G) = 0$ .

## 6 2-Cozykel und $H^2$

Sei  $G$  eine endliche Gruppe.

Sei  $M$  ein  $G$ -Modul.

Derivationen von  $G$  in  $M$  werden alternativ auch *1-Cozykel* genannt.

Im nächsten Schritt definieren wir nun *2-Cozykel*.

**Definition 26** Ein *2-Cozykel* von  $G$  in  $M$  ist eine Abbildung

$$\zeta : G \times G \rightarrow M$$

welche

$$g \cdot \zeta(g', g'') - \zeta(g \cdot g', g'') + \zeta(g, g' \cdot g'') - \zeta(g, g') = 0$$

erfüllt für  $g, g', g'' \in G$ .

Wir schreiben

$$Z^2(G, M)$$

für die Menge der 2-Cozykel von  $G$  in  $M$ .

Es ist  $Z^2(G, M)$  eine Untergruppe der abelschen Gruppe aller Abbildungen von  $G \times G$  nach  $M$ .

Insbesondere ist  $Z^2(G, M)$  eine abelsche Gruppe.

**Definition 27** Sei  $f : G \rightarrow M$  eine Abbildung.

Sei die Abbildung  $\zeta_f^{\text{inn}} : G \times G \rightarrow M$  definiert durch

$$\zeta_f^{\text{inn}}(g, g') := g \cdot f(g') - f(g \cdot g') + f(g)$$

für  $g, g' \in G$ .

Es ist  $\zeta_m^{\text{inn}}$  ein 2-Cozykel, da

$$\begin{aligned} & g \cdot \zeta_f^{\text{inn}}(g', g'') - \zeta_f^{\text{inn}}(g \cdot g', g'') + \zeta_f^{\text{inn}}(g, g' \cdot g'') - \zeta_f^{\text{inn}}(g, g') \\ = & g \cdot (g' \cdot f(g'') - f(g' \cdot g'') + f(g')) \\ & - (g \cdot g' \cdot f(g'') - f(g \cdot g' \cdot g'') + f(g \cdot g')) \\ & + (g \cdot f(g' \cdot g'') - f(g \cdot g' \cdot g'') + f(g)) \\ & - (g \cdot f(g') - f(g \cdot g') + f(g)) \\ = & 0 \end{aligned}$$

ist für  $g, g', g'' \in G$ . Ein Derivation der Form  $\zeta_f^{\text{inn}}$  für ein  $f$  heißt auch *innerer 2-Cozykel*.

Wir schreiben

$$B^2(G, M) := \{ \zeta_m^{\text{inn}} : m \in M \} \subseteq Z^2(G, M)$$

für die Menge der inneren Derivationen.

Es ist  $B^2(G, M)$  eine Untergruppe von  $Z^2(G, M)$ .

**Definition 28** Die Faktorgruppe

$$H^2(G, M) := Z^2(G, M)/B^2(G, M)$$

heißt *zweite Cohomologiegruppe* von  $G$  in  $M$ .

## 7 Zusammenhang von $H^2$ und $H^1$

Sei  $G$  eine endliche Gruppe.

Sei  $M$  ein  $G$ -Modul, der als  $\mathbb{Z}$ -Modul endlich erzeugt frei von Rang  $\text{rk } M$  sei. Wenn nötig, wird als abelsche Gruppe  $M = \mathbb{Z}^{(\text{rk } M) \times 1}$  identifiziert.

Operiere  $g \in G$  vermittelt  $\gamma(g) \in \mathbb{Z}^{(\text{rk } M) \times (\text{rk } M)}$  auf  $M$ .

**Definition 29** Wir definieren die abelsche Gruppe

$$Q := \mathbb{Q}/\mathbb{Z} .$$

Sei

$$M_Q := Q^{(\text{rk } M) \times 1} .$$

Operiere  $g \in G$  vermittelt  $\gamma(g) \in \mathbb{Z}^{(\text{rk } M) \times (\text{rk } M)}$  auf  $M_Q$ .

Es wird so  $M_Q$  zu einem  $G$ -Modul.

**Bemerkung 30** Es ist

$$H^2(G, M) \simeq H^1(G, M_Q) .$$

## 8 Berechnung von $H^2$ für Gruppen in Erzeugern und Relationen

Das Setup sei wie in §5:

Sei  $n \geq 1$ . Sei  $F$  die freie Gruppe auf  $x_1, x_2, \dots, x_n$ .

Sei  $r \geq 1$ . Seien  $w_1, \dots, w_r \in F$ .

Sei

$$G := \langle x_1, x_2, \dots, x_n : w_1, \dots, w_r \rangle .$$

Sei  $M$  ein  $G$ -Modul, der als  $\mathbb{Z}$ -Modul endlich erzeugt frei von Rang  $\text{rk } M$  sei. Wenn nötig, wird als abelsche Gruppe  $M = \mathbb{Z}^{(\text{rk } M) \times 1}$  identifiziert.

Wir wollen  $H^2(G, M) \simeq H^1(G, M_Q)$  berechnen; cf. Bemerkung 30.

**Bemerkung 31 (Skizze des Vorgehens)**

Seien

$$Z \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot n}$$

und

$$I \in \mathbb{Z}^{(\text{rk } M) \cdot n \times (\text{rk } M)}$$

wie in Bemerkung 22.

Seien wieder invertierbare Matrizen  $S \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot r}$  und  $T \in \mathbb{Z}^{(\text{rk } M) \cdot n \times (\text{rk } M) \cdot n}$  derart gefunden, daß

$$D := S \cdot Z \cdot T \in \mathbb{Z}^{(\text{rk } M) \cdot r \times (\text{rk } M) \cdot n}$$

eine Diagonalmatrix ist. Sei  $I' := T^{-1} \cdot I$ .

Dies gibt

$$\begin{array}{ccccc}
 Q^{(\text{rk } M) \times 1} & \xrightarrow{I \cdot (-)} & Q^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{Z \cdot (-)} & Q^{(\text{rk } M) \cdot r \times 1} \\
 \parallel & & \uparrow \wr T \cdot (-) & & \downarrow \wr S \cdot (-) \\
 Q^{(\text{rk } M) \times 1} & \xrightarrow{I' \cdot (-)} & Q^{(\text{rk } M) \cdot n \times 1} & \xrightarrow{D \cdot (-)} & Q^{(\text{rk } M) \cdot r \times 1}
 \end{array}$$

Es ist  $Z^1(G, M_Q)$  isomorph ersetzt durch den Kern von  $D(-)$ .

Es ist  $B^1(G, M_Q)$  isomorph ersetzt durch das Bild von  $I'(-)$ .

Wir schreiben  $b := \text{rk } D$ .

Seien die nichtverschwindenden Diagonaleinträge von  $D$  durch  $d_1, \dots, d_b \in \mathbb{Z}$  gegeben.

Der Kern von  $D(-)$  ist isomorph zu

$$\mathbb{Z}/d_1\mathbb{Z} \oplus \dots \oplus \mathbb{Z}/d_b\mathbb{Z} \oplus Q^{(\text{rk } M) \cdot n - b}$$

Aus der Theorie entnehmen wir, daß der Rang von  $I'$  gleich  $(\text{rk } M) \cdot n - b$  ist. Also ergibt sich das Bild von  $I'(-)$  zu  $Q^{(\text{rk } M) \cdot n - b}$ , welches auch auf diesen Summanden des vorstehenden Ausdrucks abgebildet wird.

Es folgt

$$H^2(G, M) \simeq H^1(G, M_Q) \simeq \mathbb{Z}/d_1\mathbb{Z} \oplus \dots \oplus \mathbb{Z}/d_b\mathbb{Z}$$

**Beispiel 32** Vorgehen wie in Beispiel 23:

```

G := SymmetricGroup(3);
G_fp<a,b>,phi := FPGroup(G);
print phi(a), phi(b); // gibt (1,2,3) und (1,2)
F<A,B> := FreeGroup(2);
rho := hom< F -> G_fp | [<A,a>,<B,b>] >; // Restklassenmorphismus

```

```

n := #Generators(G_fp);
IM_rat := IrreducibleModules(G,Rationals());
M_rat := IM_rat[3];
rk := Dimension(M_rat);
MR := MatrixRing(Integers(),rk);
rep_mat := [MR!ActionGenerator(M_rat,i) : i in [1..#Generators(G)]];
M := GModule(G,rep_mat);

prod := function(w,m)
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  return &*([MR!1] cat [rep_mat[index_seq[i]]^sign_seq[i] : i in [1..t]]) * m;
end function;

umnum := map<{-1,+1} -> {1,2} | [<1,1>, <-1,2>]>;
w := [x[1] : x in Relations(G_fp)];
r := #w;

U_elem := function(i,j,rk)
  U := RMatrixSpace(Integers(),rk,n)!0;
  U[i,j] := 1;
  return U;
end function;

derivation := function(U,w)
  w_seq := ElementToSequence(w);
  t := #w_seq;
  index_seq := [Abs(x) : x in w_seq];
  sign_seq := [Sign(x) : x in w_seq];
  elt_seq := [[ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]],
    [-rep_mat[index_seq[i]]^-1 * ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]]];
  return &+([RMatrixSpace(Integers(),rk,1)!0] cat
    [&*([MR!1] cat [rep_mat[index_seq[j]]^sign_seq[j] : j in [1..i-1]])
    * elt_seq[umnum(sign_seq[i])][i] : i in [1..#w_seq]]);
end function;

Z := RMatrixSpace(Integers(),rk*r,rk*n)!0;
for k in [1..r] do
  for i in [1..rk] do
    for j in [1..n] do
      InsertBlock(~Z,derivation(U_elem(i,j,rk),w[k]), 1 + (k-1)*rk, (i-1) * n + j);
    end for;
  end for;
end for;

```

```

print Z;
/* gibt wieder
[ 0 1 0 1]
[ 0 1 0 1]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[-2 2 1 -1]
*/

```

```

D,S,T := SmithForm(Z);
print D;
/*
[1 0 0 0]
[0 1 0 0]
[0 0 0 0]
[0 0 0 0]
[0 0 0 0]
[0 0 0 0]
[0 0 0 0]
*/

```

Folglich ist

$$H^2(G, M) \simeq H^1(G, M_Q) \simeq \mathbb{Z}/1\mathbb{Z} \oplus \mathbb{Z}/1\mathbb{Z} = 0.$$

Wir betrachten noch einen anderen Modul  $M' = MM$ .

```

MM_rat := IM_rat[2];
rk := Dimension(MM_rat);
MR := MatrixRing(Integers(),rk);
rep_mat := [MR!ActionGenerator(MM_rat,i) : i in [1..#Generators(G)]];
MM := GModule(G,rep_mat);

```

```

derivation := function(U,w)
w_seq := ElementToSequence(w);
t := #w_seq;
index_seq := [Abs(x) : x in w_seq];
sign_seq := [Sign(x) : x in w_seq];
elt_seq := [[ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]],
[-rep_mat[index_seq[i]]^-1 * ColumnSubmatrix(U,index_seq[i],1) : i in [1..t]]];
return &+([RMatrixSpace(Integers(),rk,1)!0] cat
[&*(MR!1] cat [rep_mat[index_seq[j]]^sign_seq[j] : j in [1..i-1]])
* elt_seq[umnum(sign_seq[i])][i] : i in [1..#w_seq]]);
end function;

```

```

Z := RMatrixSpace(Integers(),rk*r,rk*n)!0;
for k in [1..r] do
  for i in [1..rk] do
    for j in [1..n] do
      InsertBlock(~Z,derivation(U_elem(i,j,rk),w[k]), 1 + (k-1)*rk, (i-1) * n + j);
    end for;
  end for;
end for;
print Z;
/* gibt
[ 0 0]
[-3 0]
[ 0 0]

*/

D,S,T := SmithForm(Z);
print D;
/*
[3 0]
[0 0]
[0 0]
*/

```

Folglich ist

$$H^2(G, M') \simeq \mathbb{Z}/3\mathbb{Z} .$$

# Literatur

- [1] BOSMA, W.; CANNON, J.J.; FIEKER, C.; STEEL, A. (eds.), *Handbook of Magma functions*, Edition 2.16, 2010; cf. [magma.maths.usyd.edu.au](http://magma.maths.usyd.edu.au), [magma.maths.usyd.edu.au/calc](http://magma.maths.usyd.edu.au/calc).
- [2] EICK, B.; HOLT, D.; O'BRIEN, E., *Handbook of Computational Group Theory*, Chapman & Hall, 2005.
- [3] KÜNZER, M., *Cohomologie von Gruppen*, Skript, [pnp.mathematik.uni-stuttgart.de/lexmath/kuenzer/manuscripts.html](http://pnp.mathematik.uni-stuttgart.de/lexmath/kuenzer/manuscripts.html), 2006.
- [4] WEBER, C., *Kohomologie von Spechtmoduln*, Diplomarbeit, RWTH Aachen, 2006.
- [5] WEBER, C., *Low-degree Cohomology of Integral Specht Modules*, arxiv:0905.4209v1, 2009.
- [6] ZASSENHAUS, H., *Über einen Algorithmus zur Bestimmung der Raumgruppen*, *Comm. Math. Helv.* 21, p. 117–141, 1948.